
Agenda

e What is Data Plane Programming

e P4 — The Data Plane Programming Language
e P4 By Example

e Solving Practical Problems with P4

e Using P4

* P4 Language Evolution

What is Data Plane Programming

e What are these “planes”
e Why program the Data Plane?

Standard Telecommunications Architecture

e Traditional architecture consists Management Plane

of the three planes

e A Plane is a group of algorithms

e These algorithms
o Process different kinds of traffic
o Have different performance requirements

o Are designed using different
methodologies

o Are implemented using different
programming languages

o Run on different hardware

o

What constitutes a NOS?

e Control Plane Management Plane

e Management Plane

e How about Data Plane?

o In many modern, high-speed devices it is
no longer a part of a NOS

Designing NOS for a Fixed Data Plane

r Management Plane

Configuration CLI/GUI/SNMP...

That's how |
want to build my

network!

Control Plane

Protocol Stacks

That’s how
packets should

be processed!

That’s how |
process the
packets!

Bringing Data Plane back in the NOS

e Management Plane 3\

Configuration CLI/GUI/SNMP...

That's how | Control Plane

want to build my

network! Protocol Stacks

Benefits of Data Plane Programmability

Control and Customization. Make the device behave exactly as you want

Reliability. Reduce the risk by removing unused features

J

~

Efficiency. Reduce energy consumption and expand scale by doing only what you need

J

~

Add new features on your schedule

J

~

Telemetry. Be able to see inside the Data Plane

Exclusivity and Differentiation. No need to share your IP with the chip vendor

y CEEEEK

P4 — The Data Plane Programming Language
O
e Brief History
e Programmable Data Plane Architecture
e Typical Workflow
e P4 Concepts
P4 Architecture == Data Plane Model
Language Elements

Execution Model
Program State and Control Plane Interface

o O O O

Brief History and Trivia

e May 2013:
e July 2014:
e Aug 2014:
e Sep 2014:
e Jan 2015:
e Mar 2015:
e Nov 2016:
e May 2017:

o Apr 2016:
e Dec 2016:
e May 2017:

Initial idea and the name “P4”

First paper (SIGCOMM ACR)

First P4,, Draft Specification (v0.9.8)
P4,, Specification released (v1.0.0)
P4,,v1.0.1

P4,,v1.0.2

P4,,v1.0.3

P4,,v1.0.4

P4,; — first commits
First P4,; Draft Specification
P4,; Specification released

o Official Spelling P4 16 on terminals, P4, in publications

Copyright © 2017 — P4.org



Consortium of academic
and industry members

Open source, evolving,
domain-specific language

Permissive Apache license,
code on GitHub today

Membership is free:
contributions are welcome

Independent, set up as a
California nonprofit

The P4 Language Consortium

It's time to say "Hello Network"

P4 is a domain-specific programming language to describe the data-plane of your network.

Protocol Independent

P4 programs specify how a switch processes packets.

Target Independent
P4 is suitable for describing everything from high-

performance forwarding ASICs to software switches.

Field Reconfigurable

P4 allows network engineers to change the way their

switches process packets after they are deployed.

Copyright © 2017 — P4.org

table routing {

reads {
ipv4.dstAddr : lpm;

}

actions {
do drop;
ro;be_ipvé;

}

size: 2

}

control ingress {
apply (routing) ;

P4,. Design Goals

e Logical Evolution of P4,,
o Same basic building blocks and concepts
o More expressive and convenient to use

o More formally defined semantics
= Strong Type System

o Support for good software engineering practices
e Target-Independence
o Support for a variety of targets (ASICs, FPGAs, NICs, software)

= Language/Architecture separation
= Flexible data plane model

e Non-goals
o New constructs
o General-purpose programming

Copyright © 2017 — P4.org

P4_16 Data Plane Model

Copyright © 2017 — P4.org

>

i WA

PISA: Protocol-Independent Switch Architecture

Programmer defines the

tables and the exact Programmer declares
Programmer declares the } processing algorithm

[ how the output packet will

headers that should be look on the wire

recognized and their order
in the packet

Programmable Match-Action Pipeline
Programmable A Programmable

Parser N Deparser

~N

- = =

—m =
—m =
10 =
—m =

Match+Action
Stage (Unit)

PISA in Action

>

E; WY

e Packet is parsed into individual headers (parsed representation)

e Headers and intermediate results can be used for matching and actions
e Headers can be modified, added or removed

e Packet is deparsed (serialized)

Programmable Match-Action Pipeline
Programmable A Programmable

Parser r N\ Deparser

AL

Match+Action
Stage (Unit)

>

ii WL,

Mapping a Simple L3 Data Plane Program on PISA

> IPv4
> IPv6
Programmable Match-Action Pipeline
Programmable A Programmable

Parser 4 \ Deparser

> > g

AAAIANAAAS

HERE

Match+Action
Stage (Unit)

i WL,

Mapping a More Complex Data Plane Program on PISA

» IPv4
»|  IPv6
Programmable Match-Action Pipeline
Programmable A Programmable

Parser r \ Deparser

> >

r g

HEHH
TR

Match+Action
Stage (Unit)

P4,, Switch Model

e Ingress Pipeline
e Egress Pipeline
o Traffic Manager

o N:1 Relationships: Queueing, Congestion Control
o 1:N Relationships: Replication

o Scheduling
Implicitl
Programmable Prog;grlr?rlngble
Parser Deparser
[ B B > | —m
Packet
[ B [
1R oo | (| O
B > | || RS (| || |
- - . Scheduling -
B |EE |
[ B B> >
\ ) \ )
\

Ingress pipeline Eqgress pipeline
m g pIp g pIp .

Programmable SmartNIC Architecture (according to P4,,)

e Single, programmable pipeline

Implicitly

Programmable Programmable
Parser Deparser

B e e (.
e e (e . =
B e ) e —10
B B e —10
B B e e —m
B B ) Ee

P4,, Switch Model

e Ingress Pipeline

e Egress Pipeline

o Traffic Manager
o N:1 Relationships: Queueing, Congestion Control
o 1:N Relationships: Replication
o Scheduling

Programmable
Parser

Programmable
Deparser

Copyright © 2017 — P4.org

Alternative Switch Model

e Symmetric Ingress and Egress Pipelines

e Fully and independently programmable parsers and deparsers

\

(VAVAVAVAY)
(VAVAVAVAV

HHHH

D D

o

\
Ingress pipeline

Packet
Queuing,

Replication &
Scheduling

Egress pipeline

Extended Pipeline model

e “Non-standard” combination of programmable components
o Two parsers can be very useful for tunnel processing
o Many other combinations are also possible

e Specialized components available for advanced functionality

Programmable Programmable Programmable

Parser 1 Parser 2 eparser
B %2 B (IR EEP|  —my
= %2 2 (P> (P
B 2 D (D (> 1D
D (B 2 (| (EEE| | I
B %> D Db BB | —y
_ ElmleviE clgl =il 1=

P4 16 Approach

P4 Target An embodiment of a specific hardware implementation

. A specific set of P4-programmable components, externs, fixed components and their
P4 Architecture interfaces available to the P4 programmer
P4 Platform P4 Architecture implemented on a given P4 Target

Community-Developed

A

N —
P46 P4,; Core

Vendor-supplied

Extern Architecture
Libraries Definition

Language Library

W

P4_16 Program Portability

P4 programs C/C++ programs

Target-Independent Generally portable across a wide variety of CPUs

Architecture-Dependent Generally not portable when non-standard libraries or OS-specific
features are used

Targets can implement multiple architectures Multiple Operating Systems can be implemented for a given CPU

Multiple targets can support the same A portable Operating System or library can run on different CPUs

architecture

Community-Developed Vendor-supplied

A

A —
P4, P418 Core

D

Extern Architecture
Libraries Definition

Language iorary

W

25

V1 Architecture

e Compatible with P4_14 architecture
e Is implemented on top of bmv2-simple_switch target
o Will be gradually introduced in the due course

Programmable Programmable
Parser Deparser

\

Packet
Queuing,

Replication &
Scheduling

HHHE

\-
-
&

-

P4,. Basics

Copyright © 2017 — P4.org

P4,; Language Elements

State machine,
bitfield extraction

Parsers

_ v

Tables, Actions,
control flow
statements

Controls

Basic operations
and operators

T — w— Bistrings, headers,
Data TypeS structures, arrays '

= =
HEHEE |
AR
e =
FREE
UAVAVACAAY

CEEER

Architecture Programmable blocks
Description and their interfaces

Support for specialized

Extern Libraries components

i

P4,. Data Types

e Basic Data Types
e Derived Data Types

o Headers and metadata

Copyright © 2017 — P4.org

Simple Header Definitions

Example: Declaring L2 headers

header ethernet_t {

bit<48>

bit<48>

bit<16>
}

dstAddr;
srcAddr;
etherType;

header vlan_tag t {

bit<3>

bit<l>

bit<12>

bit<16>
}

pri;
cfi;
vid;
etherType;

struct my_headers_t {

ethernet_t

ethernet;

vlian_tag_t[2] vlan_tag;

e Basic Types
o bit<n> - Unsigned integer (bitsrting) of length n
= bit is the same as bit<1l>
o int<n> - Signed integer of length n (>=2)
o varbit<n> — Variable-length bitstring

e Derived Types

o header — Ordered collection of members
= Byte-aligned
= Can be valid or invalid
= Can contain bit<n>, int<n> and varbit<n>

o struct — Unordered collection of members
= No alignment restrictions
= Can contain any basic or derived types

o Header Stacks -- arrays of headers

Copyright © 2017 — P4.org 30



Typedef

Example: Declaring a type for MAC address e Basic Types

o bit<n> - Unsigned integer (bitsrting) of length n
= bit is the same as bit<1l>

o int<n> - Signed integer of length n (>=2)

typedef bit<48> mac_addr_t;

header ethernet_t {

mac_addr_t dstAddr; o varbit<n> — Variable-length bitstring
mac_addr_t dstAddr; e Derived Types
bit<16> etherType; o header — Ordered collection of members

= Byte-aligned
= Can be valid or invalid
= Can contain bit<n>, int<n> and varbit<n>
o struct — Unordered collection of members
= No alignment restrictions
= Can contain any basic or derived types
o Header Stacks -- arrays of headers

o typedef — An alternative name for a type

Varbit

Example: Declaring IPv4 header

typedef bit<32> ipv4_addr_t;

header ipv4_t {

bit<4>
bit<4>
bit<8>
bit<16>
bit<16>
bit<3>
bit<13>
bit<8>
bit<8>
bit<16>

ipv4_addr_t
ipv4_addr_t

by

version;
ih1;
diffserv;
totallLen;
identification;
flags;
frag0ffset;
ttl;
protocol;
hdrChecksum;
srcAddr;
dstAddr;

header ipv4_options_t {

varbit<320>

by

options;

e Basic Types
o bit<n> - Unsigned integer (bitsrting) of length n
= bit is the same as bit<1>
o int<n> - Signed integer of length n (>=2)
o varbit<n> — Variable-length bitstring
e Derived Types
o header — Ordered collection of members
= Byte-aligned
= Can be valid or invalid
= Can contain bit<n>, int<n> and varbit<n>
o struct — Unordered collection of members

= No alignment restrictions
= Can contain any derived types

o Header Stacks -- arrays of headers
o typedef — An alternative name for a type

Copyright © 2017 — P4.org

Using structs for Intrinsic Metadata

typedef bit<9> port_id_t; /*x Switch port */
typedef bit<16> mgid_t; /* Multicast Group *x/
typedef bit<5> qid_t; /* Queue 1ID */

struct parser_input_t {
port_id_t ingress_port;
bit<1> resubmit_flag;
}

struct ingress_input_t {
portid_t ingress_port;
timestamp_t ingress_timestamp;

by

struct ingress_output_t {

portid_t egress_port;
mgid_t mcast_group;
bit<1l> drop_flag;

qid_t egress_queue;

e Intrinsic Metadata is the data that a P4-
programmable components can use to
interface with the rest of the system

e These definitions come from the files,
supplied by the vendor

P4 Platform

parser_

ingress_
input_t input_t

Declaring and Initializing Variables

bit<16> my_var;
bit<8> another_var = 5;

e In P4_16 you can instantiate variables of
both base and derived types

e Variables can be initialized
© Including the composite types

ethernet t  eth; e Constant declarations make for safer code
vlan_tag_t vtag = { 3w2, 0, 12wl3, 16w0x8847 };

e Infinite width and explicit width constants
Safe constants with
explicit widths

Better than
#define!
const bit<16> ETHERTYPE_IPV4
const bit<16> ETHERTYPE_IPV6

0x0800;
0x86DD;

34

Programming the Parser

Copyright © 2017 — P4.org

Parser Model (V1 Architecture)

#include <core.p4>
#include <vimodel.p4>

/* User-defined inputs and outputs x/
struct my_headers_t {

ethernet_t ethernet;

vlan_tag_t[2] vlan_tag;

ipv4_t ipv4;

ipv6_t ipv6; The platform Initializes
b User Metadata to 0

struct my_metadata_t {
/* Nothing yet x/
}

/* System-provided inputs. Others to follow */
struct standard_metadata_t {
bit<9>  ingress_port;

+

/* Parser Declaration x/

parser MyParser(packet_in packet,
out my_headers_t hdr,
inout my_metadata_t meta,

inout standard_metada_t standard_metadata)

N

}

standard metadata

Copyright © 2017 — P4.org

MyParser

packet_in

ingress_port

Parsers in P4,,

e Parsers are special functions written in a
state machine style

e Parsers have three predefined states
o start
o accept

o reject
= Can be reached explicitly or implicitly
= What happens in reject state is defined by an architecture

e Other states are user-defined

Copyright © 2017 — P4.org

Implementing Parser State Machine

parser MyParser(packet_in packet, state parse_ipv4 {
out  my_headers_t hdr, packet.extract(hdr.ipv4);
inout my_metadata_t meta, transition select(hdr.ipv4.ihl) {
in standard_metadata_t standard_metadata) 0 .. 4: reject;
{ 5: accept;
state start { e e ipv4d options:
packet.extract(hdr.ethernet); €fault: parse_1ipva_options;
transition select(hdr.ethernet.etherType) { y
0x8100 &&& OXEFFF : parse_vlan_tag;
0x0800 : parse_ipv4; state parse_ipv4_options {
0x86DD : parse_ipv6; packet.extract(hdr.ipv4.options,
0x0806 : parse_arp; (hdr.ipv4.ihl - 5) << 2);
default : accept; transition accept;
¥ }

s

state parse_vlan_tag {
packet.extract(hdr.vlan_tag.next);
transition select(hdr.vlan_tag.last.etherType) {

0x8100 :

0x0800

state parse_ipv6 {
packet.extract(hdr.ipv6);
transition accept;

parse_vlan_tag; }

: parse_ipv4;
0x86DD :
0x0806 :

default :

parse_ipv6;
parse_arp;
accept;

Copyright © 2017 — P4.org

Selecting on Multiple fields. Parsing ARP

const bit<16> ARP_HTYPE_ETHERNET = 0x0001;
const bit<16> ARP_PTYPE_IPV4 = 0x0800;
const bit<8> ARP_HLEN_ETHERNET = 6;
const bit<8> ARP_PLEN_IPV4 = 4;

state parse_arp {
packet.extract(hdr.arp);
transition select(hdr.arp.htype,
hdr.arp.ptype,
hdr.arp.hlen,
hdr.arp.plen) {
(ARP_HTYPE_ETHERNET,
ARP_PTYPE_IPV4,
ARP_HLEN_ETHERNET,
ARP_PLEN_IPV4) : parse_arp_ipv4;
default : accept;

hdrarp =

b
hdr.arp_ipv4 =

state parse_arp_ipv4 {
packet.extract(hdr.arp_ipv4);

transition accept;

octet offset

Copyright © 2017 — P4.org

These definitions should go in

the beginning of the parser
function or in the top-level

Internet Protocol (IPv4) over Ethernet ARP packet
0 1
Hardware type (HTYPE). (Ethernet = 0x0001)
Protocol type (PTYPE) (IPv4 = 0x0800)
Hardware address length (HLEN) Protocol address length (PLEN)
Operation (OPER)
Sender hardware address (SHA) (first 2 bytes)
(next 2 bytes)
(last 2 bytes)
Sender protocol address (SPA) (first 2 bytes)
(last 2 bytes)
Target hardware address (THA) (first 2 bytes)
(next 2 bytes)
(last 2 bytes)
Target protocol address (TPA) (first 2 bytes)
(last 2 bytes)

Header verification

/* Standard errors, defined in core.p4 */

error {
NoError, //
PacketTooShort, //
NoMatch, //

StackOutOfBounds, //
OverwritingHeader, //
HeaderTooShort, //
ParserTimeout //

}

no error

not enough bits in packet for extract

match expression has no matches

reference to invalid element of a header stack
one header is extracted twice

extracting too many bits in a varbit field
parser execution time limit exceeded

/* Additional error added by the programmer x/

error { IPv4BadHeader }

state parse_ipv4 {

packet.extract(hdr.ipv4);

verify(hdr.ipv4.version == 4, error.IPv4BadHeader);
verify(hdr.ipv4.ihl >= 5, error.IPv4BadHeader);
transition select(hdr.ipv4.ihl) {

5: accept;

default: parse_ipv4_options;

Copyright © 2017 — P4.org

Lookahead

Example: Typical MPLS Heuristic * lookahead is a generic method

header ip46_t { /* Common for both IPv4 and IPv6 x*/ © Compller ge.nerates the methOd. Wlt.h the [Prefp=r
bit<d> version: return type (ipv4_t) at the compile time
bit<4> reserved; e Returns the packet data without advancing

4 the cursor

ST (LT o The packet length is c:‘,till checked .
packet.extract(hdr.mpls.next); e Much safer and easier to use than bit

transition select(hdr.mpls.last.bos) {
0: parse_mpls;
1: guess_mpls_payload;

offsets

b

state guess_mpls_payload {
transition select(packet.lookahead<ip46_t>().version) {
4 : parse_inner_ipv4;
6 : parse_inner_ipv6;
default : parse_inner_ethernet;

Copyright © 2017 — P4.org 41



Programming the Match-Action Pipeline

e Controls
e Actions
e Tables

Copyright © 2017 — P4.org



Controls in P4

e Very similar to C functions without loops
o Algorithms should be representable as Direct Acyclic Graphs (DAG)

e Represent all kinds of processing that are expressible as DAG:
o Match-Action Pipelines
o Deparsers
o Additional processing (checksum updates)

o Interface with other blocks via user- and architecture-defined
data

Copyright © 2017 — P4.org



Simple Reflector (V1 Architecture)

control MyIngress(
inout my_headers_t hdr,
inout my_metadata_t meta,
inout standard_metadata_t standard_metadata)
{
bit<48> tmp;
apply {
tmp = hdr.ethernet.dstAddr;
hdr.ethernet.dstAddr = hdr.ethernet.srcAddr;
hdr.ethernet.srcAddr = tmp;
standard_metadata.egress_spec = ©
standard_metadata.ingress_port; {g
©
J T
¥ £
| -
2
@
o
=
8
7]

Copyright © 2017 — P4.org

ingress_port egress_spec

mcast_grp

resubmit_flag

. . clone_spec
ingress_timestamp

resubmit_flag

egress_queue

T

elepejow pJepue)s

How does it work? (V1 Architecture)

Mylngress()

hdr hdr

ingress_port egress_spec

resubmit_flag mcast_grp

. . clone_spec
ingress_timestamp

resubmit_flag

egress_queue

MyEgress()

hdr

Packet
Re p I | Catl O n 4 egress_port

Engine ll!a{ egress._rid
clone_id clone_id

packet_length

e o drop_flag

16
deq_qgdepth

Simple Actions and Expressions

const bit<9> DROP_PORT = 511; /* Specific to V1 architecture x/

action mark_to_drop() { /* Already defined in vlmodel.p4 x/
standard_metadata.egress_spec = DROP_PORT;
standard_metadata.mgast_grp = 0;

}

control MyIngress(inout my_headers_t hdr,

inout my_metadata_t meta,
inout standard_metadata_t standard_metadata)
{

/* Local Declarations */

action swap_mac(inout bit<48> dst, inout bit<48> src) {
bit<48> tmp;
tmp = dst; dst = src; src = tmp;

s

action reflect_to_other_port() {
standard_metadata.egress_spec =
standard_metadata.ingress_port ~ 1;

e Very similar to C functions

e Can be declared inside a control or
globally

o Parameters have type and direction
o Variables can be instantiated inside

o Standard Arithmetic and Logical
operations are supported

*
o+, -
e /AN
o~ &, |, A >>, <<
o== 1= > >= < <=

o No division/modulo
o Additional operations:
o Bit-slicing: [m:]
= Works as I|-value too
o Bit Concatenation: ++

Copyright © 2017 — P4.org

Simple Actions and Expressions

const bit<9> DROP_PORT = 511; /% V1 Architecure-specific */ /* The body of the control */
action mark_to_drop() { /x Already defined in vimodel.p4 x/ apply {
standard_metadata.egress_spec = DROP_PORT; if (hdr.ethernet.dstAddr[40:40] == 0x1) {
standard_metadata.mgast_grp = 0; mark_to_drop();
} } else {
swap_mac(hdr.ethernet.dstAddr,
control MyIngress(inout my_headers_t hdr, hdr.ethernet.srcAddr);
inout my_metadata_t meta, reflect_to_other_port();
inout standard_metadata_t standard_metadata) }
{ }
/* Local Declarations */ }

action swap_mac(inout bit<48> dst, inout bit<48> src) {
bit<48> tmp;
tmp = dst; dst = src; src = tmp;

action reflect_to_other_port() {
standard_metadata.egress_spec =
standard_metadata.ingress_port ~ 1;

Actions Galore: Operating on Headers

Example: Encapsulating IPv4 into a 2-label MPLS packet e Header Validity bit manipulation:
header mpls_t { .
bit<2os label; o header.setValid() — add_header
bit<3> exp; o header.setInvalid() — remove_header
bit<1l> bos; . .
bit<g» ttls o header.isValid()
} e Header Assignment

O
action ipv4_in_mpls(in bit<20> labell, in bit<20> label2) { From tuples

hdr.mpls[0].setValid();

hdr.mpls[0].label = labell;

hdr.mpls[@].exp = 0;

hdr.mpls[@].bos = 0;

hdr_mpls[0].ttl = 64;

hdr.mpls[1].setValid(); if() statements
hdr.mpls[1] = { label2, @, 1, 128 }; are allowed in

actions too!

if (hdr.vlan_tag.isValid()) {
hdr.vlan_tag.etherType = 0x8847;
} else {
hdr.ethernet.etherType = 0x8847;
}

Copyright © 2017 — P4.org

Actions Galore: Operating on Headers

action decap_ip_ip() { o Header Validity bit manipulation:
hdr.ipv4 = hdr.inner_ipv4; .
hdr.inner_ipv4.setInvalid(); o header.setValid() — add_header
) o header.setInvalid() — remove_header
action pop_mpls_label() { o header.isValid()
hdr.mpls.pop_front(1); ¢ Header Assignment
¥
oheader = { f1, f2, ..., fn }
action push_mpls_label(in bit<20> label, in bit<3> exp) { o headerl = header?2
hdr.mpls. h_f t(1); . .
har moleloll setvatid () « Special operations on Header Stacks
hdr.mpls[@0] = { label, exp, 0, 64 }; o In the parsers
} » header_stack.next

= header_stack.last
= header_stack.lastIndex

o In the controls
= header_stack[i]
= header_stack.size
= header_stack.push_front(int count)
= header_stack.pop_front(int count)

Actions Galore: Bit Manipulation

Example: Forming Ethernet MAC address for IPv4 Multicast e Special Operations for bit manipulation:
Packets . : .
o Bit-string concatenation

action set_ipmcv4_mac_da_1() { ) Bit-slicing
hdr.ethernet.dstAddr = 24w@x01005E ++ 1w ++
hdr.ipv4.dstAddr[22:0];

}

action set_ipmcv4_mac_da_2() {
hdr.ethernet.dstAddr[47:24] = 0x01005E;
hdr.ethernet.dstAddr[23:23] = 0;;

hdr.ethernet.dstAddr[22:0]

hdr.ipv4.dstAddr[22:0];

Match-Action Tables

e The fundamental units of the Match-Action Pipeline
o What to match on and match type
o Alist of possible actions
o Additional properties

= Size

= Default Action
= Entries

= etc.

o Each table contains one or more entries (rows)

e An entry contains:
o A specific key to match on

o A single action
= to be executed when a packet matches the entry

o (Optional) action data

Tables: Match-Action Processing

Contril Plane

Headers and Metadata

Lookup Ke

Siajaweled
(sue|dereq)
|[euonoalqg

m Action Data

10)09]8S
SSIN/HH

Directionless
(Action Data)
Parameters

Headers and . Headers and
Metadata iﬂ?:rl\t DefalIJDI;tA;ctlon Action Metadata
(Input) Execution (Output)

o

Example: Basic IPv4 Forwarding

192.168.1.1 192.168.23.45
\

[

[ ——

192.168.1.2

Ry

|

_1 192.168.23.254

192.168.1.3 10.1.2.0/22

|

[ _—

192.168.1.254]

192.168.1.4

[

||

| key | Action Action Data

192.168.1.1 13_switch

port= mac_da= mac_sa= vlan=

192.168.1.2 13_switch port=" mac_da= mac_sa= vlan=...
192.168.1.3 I3_drop

192.168.1.254 13_12_switch port=

192.168.1.0/24 I3_12_switch port=

10.1.2.0/22 13_switch_ecmp ecmp_group=

e Data Plane (P4) Program

o Defines the format of the table
= Key Fields
= Actions
= Action Data

o Performs the lookup
o Executes the chosen action

e Control Plane (IP stack,

Routing protocols)

o Populates table entries with

specific information

= Based on the configuration

= Based on automatic discovery
» Based on protocol calculations

Defining Actions for L3 forwarding

action 13_switch(bit<9> port, e Actions can use two types of
bit<48> new_mac_da,
bit<48> new_mac_sa, parameters
. bit<12> new_vlan) o Directional (from the Data Plane)
/* Forward the packet to the specified port x/ o Directionless (from the Control
standard_metadata.metadata.egress_spec = port; Plane)
P e Actions thgt are called directly:
hdr.ethernet.dstAddr = new_mac_da; o Only use directional parameters
hdr.ethernet.srcAddr. = mac_sa; ° ACtiOI'IS used in tables:
hdr.vlan_tag[0].vlanid = new_vlan; _ _ _
o Typically use direction-less
/* IP header modification (TTL decrement) x/ parameters
hdr.ipv4.ttl = hdr.ipv4.ttl — 1; . . .
S P P o May sometimes use directional

parameters too
action 13_12_switch(bit<9> port) {
standard_metadata.metadata.egress_spec = port;

}

action 13_drop() {
mark_to_drop();
}

Defining Actions

const bit<l> NEXTHOP_TYPE_SIMPLE = 0;
const bit<l> NEXTHOP_TYPE_ECMP = 1;

struct 13_metadata_t {
bit<l> nexthop_type;
bit<16> nexthop_index;

}

action 13_switch_nexthop(out 13_metadata_t 13_meta,
bit<16> nexthop_index)

{

13_meta.nexthop_type
13_meta.nexthop_index

NEXTHOP_TYPE_SIMPLE;
nexthop_index;

}

action 13_switch_ecmp(out 13_metadata_t 13_meta,
bit<12> ecmp_group)

{

13_meta.nexthop_type
13_meta.nexthop_index

NEXTHOP_TYPE_ECMP;
(bit<16>)ecmp_group;

P4_16 has

typecasts too!

e Actions can use two types of
parameters
o Directional (from the Data Plane)

o Directionless (from the Control
Plane)

o Actions that are called directly:

o Only use directional parameters
o Actions used in tables:

o Typically use direction-less
parameters

o May sometimes use directional
parameters too

= |f present, directional parameters
always go first

Match-Action Table

Example: A typical L3 (IPv4) Host table

table ipv4_host {

key = {
meta.ingress_metadata.vrf
hdr.ipv4.dstAddr

b

actions = {
13_switch; 13_12_switch;
13_drop; noAction;

I

default_action = noAction();

size = 65536;

(Exact Match)

These are the only
possible actions. Each

particular entry can have
only ONE of them.

/* Defined in core.p4 */

action noAction() { }

1 192.168.1.10

100 192.168.1.10

1 192.168.1.3
DEFAULT

13_switch port_id= mac_da= mac_sa=
13_I2_switch port_id=<CPU>

I13_drop

noAction

Copyright © 2017 — P4.org

Match-Action Table (Longest Prefix Match)

Example: A typical L3 (IPv4) Routing table

Different fields can
table ipv4_lpm { use different

key = { match types
meta.ingress_metadata.vrf : exact;

hdr.ipv4.dstAddr : lpm;
} Different fields can
actions = { use different
13_switch; match types

13_12_switch;
13_switch_nexthop(meta.13.nexthop_info);
13_switch_ecmp(meta.13.nexthop_info);
13_drop; noAction;

} Prefix length also
const default_action = 13_12_switch(CPU_PORT); serves as a priority
size = 16384; indicator

1 192.168.1.0 /24 13_I2_switch port_id=3

10 10.0.16.0/ 22 13_ecmp ecmp_index=12

1 192.168.0.0 / 16 13_switch_nexthop nexthop_index=451
1 0.0.0.0/0 13_switch_nexthop nexthop_index=1

- DEFAULT | I3_I2_switch port_id=CPU_PORT
m Copyright © 2017 — P4.org

Match-Action Table (Ternary Match)

Example: A more powerful L3 (IPv4) Routing table

table ipv4_1lpm {

key = {
meta.ingress_metadata.vrf : ternary;
hdr.ipv4.dstAddr : ternary;
¥
actions = {
13_switch;

13_12_switch;
13_switch_nexthop(meta.13.nexthop_info);
13_switch_ecmp(meta.13.nexthop_info);
13_drop;
noAction;
¥
const default_action = 13_12_switch(CPU_PORT);
size = 16384;

Explicitly Specified Priority

100 0x001/0xFFF  192.168.1.5 / 255.255.255.255
10 0x000/0x000  192.168.2.0/255.255.255.0

10 0x000/0x000  192.168.3.0/255.255.255.0

5 0x000/0x000  0.0.0.0/0.0.0.0

13_swith_nexthop nexthop_index=10
I3_switch_ecmp ecmp_index=25
I13_switch_nexthop nexthop_index=31

13_12_switch port_id=64

58



Using Tables in the Controls

control MyIngress(inout my_headers_t hdr, . Declare ACtiOﬂS
inout my_metadata_t meta, . L Lo
inout standard_metadata_t standard_metadata) o Declaration is instantiation
{ | e Declare Tables
/* Declarations */ . .. N
action 13_switch(...) { . . .} o Declaration is instantiation
action 13_12_SW1tCh(-..) { 0O o o } ° Apply() Tables —_ Perform Match_Action
table assign_vrf { . . . } o Make sure the table matches on valid headers
table ipv4_host { . . . }
table ipv6_host { . . . }
/* Code */
apply {

assign_vrf.apply();

if (hdr.ipv4.isValid()) {
ipv4_host.apply();

}

Using the Match Results

apply {

if (hdr.ipv4.isValid()) {
if (!ipv4_host.apply().hit) {
ipv4_lpm.apply();

b
b
¥
apply {
switch (ipv4_lpm.apply().action_run) {
13_switch_nexthop: { nexthop.apply(); }
13_switch_ecmp: { ecmp.apply(); }
13_drop: { exit; }
default: { /* Not needed. Do nothing
¥
}

o Apply method returns a special result:
o A boolean, representing the hit
o An enum, representing a selected action

e Switch() statement
o Only used for the results of match-action
o Each case should be a block statement

o Default case is optional
= Means “any action” not “default action”

o o Exit and Return Statements
o return — go to the end of the current control
o exit — go to the end of the top-level control
o Useful to skip further processing

Match Kinds (types)

/* core.pd4 x/

match_kind {
exact,
ternary,
Tpm

}

/* vlmodel.p4d x/

match_kind { /* Augments the standard definition }
range,
selector

}

/* Some other architecture x/

match_kind {
regexp,
range,
fuzzy,
telepathy

e match_kind is a special type in P4

e core.p4 defines three basic match kinds
o Exact match
o Ternary match
o LPM match

e Architectures can add their own match
kinds

Advanced Matching

table classify_ethernet {  Tables keys can be arbitrary expressions

key = {
hdr.ethernet.dstAddr : ternary; e Check header Valldlty
hdr.ethernet.srcAddr : ternary; C :
hdr.vlan_tag[0].isValid() : ternary; e header'_lsvalld( ) .
hdr.vlan_tag[1].isValid() : ternary; e Use only important bits

 one . o header.field[msb: lsb]
nalformed_ethernet; = hdr.ipv6.dstAddr[127:64] : 1lpm;
unicast_untagged; e Fantasy is your limit:
unicast_single_tagged; ohdr.ethernet.srcAddr ~
unicast_double_tagged;

multicast_single_tagged;
multicast_double_tagged;
broadcast_untagged;

broadcast_single_tagged;
broadcast_double_tagged;

Table Initialization

/* Continued from previous slide */
const entries = {
/* { dstAddr, srcAddr, vlan_tag[@].isValid(), vlan_tagl[1].isValid() } : action([action_datal) */

{ 48w000000000000, _ ~, _ } : malformed_ethernet(ETHERNET_ZERO DA);
{ _, 48w000000000000, _, _ } : malformed_ethernet(ETHERNET_ZERO_SA);
{ _, 48w010000000000 &&& 48w010000000000, _, _ } : malformed_ethernet(ETHERNET_MCAST_SA);
{ 48wFFFFFFFFFFFF, _, 0, _ } : broadcast_untagged();

{ 48wFFFFFFFFFFFF, ., 1, @ } : broadcast_single_tagged();

{ 48wFFFFFFFFFFFF, , 1, 1 } : broadcast_double_tagged();

{ 48w010000000000 &&& 48w010000000000, _, 0, _ } : multicast_untagged();

{ 48w010000000000 &&& 48w010000000000, _, 1, 0 } : multicast_single_tagged();

{ 48w010000000000 &8& 48w010000000000, _, 1, 1 } : multicast_double_tagged();

{_, , 0, _ } : unicast_untagged();

{_, , 1, @ } : unicast_single_tagged();

{ _, ., 1, 1 } : unicast_double_tagged();

Packet Deparsing

Deparsing

control MyDeparser(packet_out packet,
in my_headers_t hdr)

{

apply {
/* Layer 2 x/
packet.emit(hdr.
packet.emit(hdr.

/* Layer 2.5 x/
packet.emit(hdr.

/* Layer 3 x/
/* ARP x/
packet.emit(hdr.
packet.emit(hdr.
/x IPv4 x/
packet.emit (hdr.
/* IPvV6 */
packet.emit(hdr.

/*x Layer 4 %/

packet.emit(hdr.
packet.emit(hdr.
packet.emit (hdr.

ethernet);
vlan_tag);

mpls);

arp);
arp_ipv4);

ipv4);
ipv6);
icmp);

tcp);
udp) ;

o Assembling the packet from headers
e Expressed as another control function
o No need for another construct
e packet_out — defined in core.p4
o emit(header) — serialize the header if it is valid

o emit(header_stack) — serialize the valid elements
in order

o Advantages:
o Decoupling of parsing and deparsing

Simplified Deparsing

struct my_headers_t {

ethernet_t ethernet;

vlian_tag_t [2] vlan_tag;

mpls_t [5] mpls;

arp_t arp;

arp_ipv4_t arp_ipv4;

ipv4_t ipvé4;

ipve_t ipv6;

icmp_t icmp;

tep_t tcp;

udp_t udp;
¥
control MyDeparser(packet_out packet,

in my_headers_t hdr)

{

apply {

packet.emit(hdr);

}

}

e Simply keep the header struct organized
o Headers will be deparsed in order

Externs

The Need for Externs

e Most platforms contain specialized facilities
o They differ from vendor to vendor

o They can’t be expressed in the core language
= Specialized computations

o They might have control-plane accessible state or configuration
e The language should stay the same

o In P4,, almost 1/3 of all the constructs were dedicated to specialized
processing

o In P44 all specialized objects use the same interface
e Objects can be used even if their implementation is hidden
o Through instantiation and method calling

Stateless and Stateful Objects

o Stateless Objects: Reinitialized for each packet

o Variables (metadata), packet headers, packet_in, packet_out

o Stateful Objects: Keep their state between packets
o Tables

o Externs
= P4,,: Counters, Meters, Registers, Parser Value Sets, Selectors, etc.

Parser Value Set

Table

Counter
Meter

Register

apply()
get()

read()

Modify/Write State

count()
execute ()

write()

Configuration Only

Modify/Write

Yes*
Configuration Only

Yes

Counters in V1 Architecture

AR R B LU B o Extern definition contains

enum CounterType { o The instantiation method
packets, = Has the same name as the extern
bytes, = |s evaluated at compile-time

e sl Tyt
N o Methods to access the extern

= Very similar to actions
/*x An array of counters of a given type */ = Can return values too
extern counter {
counter(bit<32> instance_count, CounterType type);
void count(in bit<32> index); e Enums in P416
y o Abstract values

= No specific (numerical representation)

b

Copyright © 2017 — P4.org 70



Using the V1 Architecture Counters

control MyIngress(inout my_headers_t hdr, . . .
inout my metadata. t Teta) » Instantiate an extern inside the control
inout standard_metadata_t standard_metadata) o Call the instantiation method

{ = Parameters must be known at compile-time

counter(8192, CounterType.packets_and_bytes) ingress_bd_stats; y .
e Use extern’s methods in actions or

action set_bd(bit<16> bd, bit<13> bd_stat_index) { .
meta.12.bd = bd; directly
ingress_bd_stats.count((bit<32>)bd_stat_index);
ks
table port_vlan {
key = {
standard_metadata.ingress_port : ternary; counter
hdr.vlan_tag[@].isvalid() : ternary; ingress_bd_stats
hdr.vlan_tag[@].vid : ternary; 0
¥
actions = { set_bd; mark_for_drop; } T T
default_action = mark_for_drop();
}
table port_vlan
Ke Action Action Data 8191
apply { ABCD_0123 set_bd
port_vlan.apply(); set_bd i
bd [bd stat_index A
e set_bd
} set_bd
1 set_bd
BASE_F007 set_bd

71

Meters in V1 Architecture

Definition
/* Definition in vlmodel.p4 x*/
enum MeterType {

packets,
bytes

extern meter {

meter(bit<32> instance_count, MeterType type);

void execute_meter<T>(in bit<32> index, out T result);

This is a template

definition. The method
will accept the
parameter of any type

Color Coding

0 — Green

2 -- Red

Usage

typedef bit<2> meter_color_t;

const meter_color_t METER_COLOR_GREEN = 0;
const meter_color_t METER_COLOR_YELLOW = 1;
const meter_color_t METER_COLOR_RED = 2;

meter(1024, MeterType.bytes) acl_meter;

action color_my_packets(bit<10> index) {
acl_meter.execute_meter((bit<32>)index,

}
table acl {
key = { . . .}
actions = { color_my _packets; . . .}
¥
apply {
acl.apply();
if (meta.color == METER_COLOR_RED) {
mark_to_drop();
}
¥

Registers in V1 Architecture

Definition
/* Definition in vlmodel.p4 x*/

extern register<T> {
register(bit<32> instance_count);
void read(out T result, in bit<32> index);
void write(in bit<32> index, in T value);

Usage (Calculating Inter-Packet Gap)
register<bit<48>>(16384) last_seen;

action get_inter_packet_gap(out bit<48> interval,
bit<14> flow_id)
{
bit<48> last_pkt_ts;

/* Get the time the previous packet was seen *x/
last_seen.read((bit<32>)flow_id,
last_pkt_ts);

/* Calculate the time interval x/
interval = standard_metadata.ingress_global_timestamp —
last_pkt_ts;

/* Update the register with the new timestamp x/

last_seen.write((bit<32>)flow_id,
standard_metadata.ingress_global_timestamp);

Direct Counters — Extending Tables with Externs

Definition Usage (Implementing ACL Counters)

/* Definition in vlmodel.p4 */ direct_counter(CounterType.packets_and_bytes) acl_counters;

extern direct_counter {

direct_counter(CounterType type);

void count();

action acl_deny() {

mark_to_drop();
acl_counter.count();

Possible, because

of the counter
attribute in the table

} b
table ip_acl {
key = {

ipv4_metadata. lkp_ipv4_sa : ternary;

ipv4_metadata. lkp_ipv4_da : ternary;

13_metadata. lkp_ip_proto : ternary;
table ip_acl counters = acl_counter 13_metadata.lkp_14_sport : ternary;
Match Fields Action Sel Action Data Counter 13_metadata.lkp_l4_dport : ternary;
ABCD_xxxx_0123 acl_deny counter A

by

8b[ 8b pkt/byte counts actions = {
I acl permit: Ta_bles can havge'
| _p H architecture-specific
acl_deny; properties
BABE_F007_XxXxxx counter Z 1
size = 4096;

counters = acl_counters;

Hashes and Random Numbers — Stateless Externs

Definition
/* Definition in vlmodel.p4 x*/

enum HashAlgorithm {

crc32,
crc32_custom,
crcle,
crcl6e_custom,
random,
identity
}
extern void hash<T, D>(out T result,
in HashAlgorithm algo,
in T base,
in D data,
in T max) ;

extern void random<T>(out T result, in T lo, in T hi);

extern Checksuml6 {
Checksum16() ;
bit<16> get<D>(in D data);
}

Usage (Computing a Flow Hash)

bit<10> flow_hash;

if (hdr.ipv4.isValid()) {
hash(flow_hash, HashAlgorithm.crc32,

10w0,

{ hdr.ethernet.dstAddr, hdr.ethernet.srcAddr,
hdr.ipv4.srcAddr, hdr.ipv4.dstAddr,
hdr.ipv4.protocol },

10w1023);

} else if (hdr.ipv6.isValid()) {
hash(flow_hash, HashAlgorithm.crc32_custom,

10w0,

{ hdr.ethernet.dstAddr, hdr.ethernet.srcAddr,
hdr.ipv6.srcAddr, hdr.ipv6.dstAddr,
hdr.ipv6.nextHdr },

10w1023);

} else {
hash(flow_hash, HashAlgorithm.crcl6,

10w0,

{ hdr.ethernet.dstAddr, hdr.ethernet.srcAddr,
hdr.ethernet.etherType },

10w1023);

Header Checksums (V1 Architecture)

IPv4 Checksum Verification

control MyVerifyChecksum(in my_headers_t hdr,
inout my_metadata_t meta)

{
Checksum16() ipv4_checksum;
bit<16> ck;
apply {
if (hdr.ipv4.isValid()) {
ck = ipv4_checksum.get(
{
hdr.ipv4.version, hdr.ipv4.ihl,
hdr.ipv4.diffserv, hdr.ipv4.totallLen,
hdr.ipv4.identification, hdr.ipv4.flags,
hdr.ipv4.fragOffset, hdr.ipv4.ttl,
hdr.ipv4.protocol, hdr.ipv4.srcAddr,
hdr.ipv4.dstAddr
});
if (hdr.ipv4.hdrChecksum != ck) {
mark_to_drop();
¥
b
b
b

IPv4 Checksum Update

control MyComputeChecksum(inout my_headers_t hdr,
inout my_metadata_t meta)

{
Checksum16() ipv4_checksum;
apply {
if (hdr.ipv4.isValid()) {
hdr.ipv4.hdrChecksum = ipv4_checksum.get(
{
hdr.ipv4.version, hdr.ipv4.ihl,
hdr.ipv4.diffserv, hdr.ipv4.totallLen,
hdr.ipv4.identification, hdr.ipv4.flag,
hdr.ipv4.fragOffset, hdr.ipv4.ttl,
hdr.ipv4.protocol, hdr.ipv4.srcAddr,
hdr.ipv4.dstAddr
i
}
b
}

Assembling the whole program

P4 Program Structure

* P4 programs are compiled as a single module
o No linking of separate modules (so far)
e P4 compiler passes the program through C Preprocessor first

o Use #include to modularize the code
o P4 specification mandates only a subset of CPP features to be supported

e Objects must be defined before they can be used

Overall P4 Program Structure

/* —%— P4_16 —*— %/
#include <core.p4>
#include <vlmodel.p4>

/xkkkskokskkskskokskkkkkkkkkkk T Y P E S skkokskokskskokskokskokokskokskekokskokskkkk /

typedef bit<48> mac_addr_t;
header ethernet_t { /% Slide 30 %/ }
struct my headers_t { /% Slide 30 %/ }

/kxkkkkkckkckkkkkkkk C O N S T AN T S skkkskkskskskskskskkskskskskskskskskk /
const mac_addr_t BROADCAST_MAC = OxFFFFFFFFFFFF;

/*xxxkkkkkkk P A RS ER S and
parser MyParser(...)
control MyVerifyChecksum(...)

C T R O L S skkkekkskkkskkk/

{

{
control MyIngress(...) {

{.

{7/

{7/

0N
/* Slide 38 x/ }
/* Slide 76 *x/ }
/* Slide 44 x/ }
- ¥
Slide 76 x/ }
* Slide 65 *x/ }

control MyEgress(...)
control MyComputeChecksum(...)
control MyDeparser(...)

/xkkkrkkkkkkkkkk F U L L P ACK A G E skkkkkkkkkrkkkkkkk/
V1Switch(
MyParser(), MyVerifyChecksum(),
MyEgress(), MyComputeChecksum(),
) main;

MyIngress(),
MyDeparser()

o Start with Emacs-style comment to select
the proper editor mode

Include the core library

Include the architecture-specific file(s)
Define Types

o typedefs, headers, structs, ...

Define Constants

Define Parsers and Controls

Assemble the top-level controls in a
package

o Package is defined by the Architecture

= Represents the set of programmable P4 components
and their interfaces

o The name of the package must be main
e That’s it!

Defining Your Own Controls

L3 Processing Function

control process_13(in my_headers_t hdr,
inout 13_metadata_t 13_meta)

{

/* Local variables for L3 processing */

/* Tables and actions for L3 Processing x/
table ipv4_host { . . . }
table ipv4_1lpm { . . . }
table ipv6_lpm { . . . }

apply {
if (13_meta.do_ipv4) {
ipv4_host.apply();
} else {
ipv6_lpm.apply();

Top-Level Control

struct 13_metadata_t {

bool do_ipv4;
nexthop_id_t nexthop;
}
control MyIngress(inout my_headers_t hdr,
inout my_metadata_t meta,
inout standard_metadata_t standard_metadata)
{
13_metadata_t 13_meta;
apply {
if (hdr.ipv4.isValid() || hdr.ipv6.isValid()) {
13_meta.do_ipv4 = hdr.ipv4.isValid();
process_13.apply(hdr, 13_meta);
}
}
1 apply() method applies

to controls too

Working with P4

Basic Workflow

Program-independent
CLI and Client

TCP Socket

(Thrift)
l Program-independent

Control Server

i p4c-bm2-ss
@ii EE‘

l P4
: Debugger
test.json

Packet Packet
generator : sniffer

Linux Kernel

Step 1: P4 Program Compilation

test.p4 $ pdc-bm2-ss -0 test.json test.p4

l

test.json

p4c -bm2-ss

Step 2: Preparing veth Interfaces

$ sudo ~/p4lang/tutorials/examples/veth setup.sh

# ip link add name vethO type veth peer name vethl

# for iface in “wvethO vethl”; do
ip link set dev ${iface} up
sysctl net.ipvé6.conf.S${iface}.disable ipv6=1
TOE OPTIONS="rx tx sg tso ufo gso gro lro rxvlan txvlan rxhash”
for TOE OPTION in $TOE OPTIONS; do

/sbin/ethtool --offload $intf "S$STOE OPTION”
done
done

test.json

84

Copyright © 2017 — P4.org



Step 3: Starting the model

$ sudo simple switch --log-console --dump-packet-data 64 \
-i 0@veth0 -i 1Q@veth2 .. \
test.json

TCP Socket
(Thrift)

Program-independent
Control Server

U

veth0.pcap

Linux Kernel

Step 4: Starting the CLI

$ simple switch CLI

Program-independent
CLI and Client
TCP Socket
(Thrift)

Program-independent
Control Server

test.json

Linux Kernel

Step 5: Sending and Receiving Packets

Program-independent
Control Server

* scapy
p = Ethernet()/IP()/UDP()/"Payload”
sendp(p, iface="veth0”)

 Ethereal, etc..

* scapy
sniff(iface="veth9”, prn=lambda x: x.show())

* Wirehark, tshark, tcpdump

Packet l Packet
Generator Sniffer

Linux
Kernel

Working with Tables in simple_switch_CLI

RuntimeCmd: show_tables
m filter [meta.meter tag(exact, 32)]
m table [ethernet.srcAddr (ternary, 48)]

RuntimeCmd: table info m_table
m table [ethernet.srcAddr (ternary, 48)]

KAk hkhkhhk kA hAAAA A A A AR A A Ak bk hhhh kA A ARk Ak A A A A A Ak dhkhhhkhhh kA kA kA dkhkddrhdrhdhhkhkhkhkhkhkhkhk Ak, ,k k%%

nop

[IJm action [meter idx(32)]

RuntimeCmd: dump_table m_table Value and mask for ternary

m table: matching. No spaces around
“&&&”

0: aaaaaaaaaaaa &&& fEfffffFfFffff => m action - O,
SUCCESS

RuntimeCmd: table_add m table m action 01:00:00:00:00:00&&&01:00:00:00:00:00

Adding entry to ternary match table m table “=>" separates the

match key: TERNARY-01:00:00:00:00:00 &&& 01:00:00:00:00:00 key from the action
action: m_action o
runtime data: 00:00:00:05

SUCCESS

entry has been added with handle 1

All subsequent

RuntimeCmd: table delete 1 operations use the

entry handle
o

Conclusions

Why P4,,?

o Clearly defined semantics

o You can describe what your data plane program is doing
e Expressive

o Supports a wide range of architectures through standard methodology
e High-level, Target-independent

o Uses conventional constructs

o Compiler manages the resources and deals with the hardware
e Type-safe

o Enforces good software design practices and eliminates “stupid” bugs
o Agility

o High-speed networking devices become as flexible as any software
* Insight

o Freely mixing packet headers and intermediate results

Things we covered

e The P4 "world view”
o Protocol-Independent Packet Processing
o Language/Architecture Separation
o If you can interface with it, it can be used

e Main Data Types

o Constructs for packet parsing
o State machine-style programming

e Constructs for packet processing
o Actions, tables and controls

e Packet deparsing
e Basic Program Structure

Things we didn’t cover

e Mechanisms for modularity
o Calling other parsers or controls from top-level ones
e Details of variable-length field processing
o Parsing and deparsing of options and TLVs
o Architecture definition constructs
o How these “templated” definitions are created
e Many features of V1 architecture
o How to do learning, multicast, cloning, resubmitting
e Architectures other than V1
o Latest constructs
o Header unions
e Control plane interface

Thank you

